

Методические указания для практикума по обработке дисперсионных спектров в пакете MIDAS

Для начала работы в пакете MIDAS служит команда:

```
inmidas
```

либо, если требуется продолжение предыдущей сессии MIDAS

```
gomidas
```

Внутри MIDAS можно получить информацию по командам и их аргументам с помощью команды help, например:


```
help indisk/fits
```

Для конвертации спектра из FITS-формата во внутренний формат MIDAS служит команда:

```
indisk/fits obj1.fits obj1
```

Загрузить файл в окно просмотра:

```
loa/ima obj1
```



В созданном окне появляется ПЗС кадр со спектром звезды. Яркие точки на темном фоне — следы космических частиц. Яркость кадра при первом выводе настраивается автоматически. В дальнейшем команду `loa/ima` можно дополнить необязательным аргументом `cuts=min,max`, которым устанавливаются минимальное и максимальное значения интенсивности выводимого на экран изображения. При первом выводе на экран `min=average-3sigma`, `max=average+3sigma`, где `average` и `sigma` — среднее и дисперсия распределения интенсивности по кадру. При `cuts=0,0` или `cuts=0` Midas автоматически приводит эти значения к минимальному и максимальному значению в кадре. Также можно изменять масштаб по X и Y аргументом `scale=`. При отрицательном значении будет уменьшение выводимого изображения. Более подробно смотрите с помощью

встроенного помощника: help loa/ima Очистить окно изображений можно командой cle/dis

```
loa/ima obj1 cuts=0,0 scale=-3
```

Команда: get/cur

позволяет с помощью курсора определить координаты выбранных точек и их интенсивность на изображении. Команда никак не изменяет изображения. Щелкните левой клавишей мышки по центру спектра. В терминале появится информация подобная этой:

```
frame: obj1.bdf (data = uI2)
      plane_no 1 loaded
cursor #0
      frame pixels      world coords      intensity
      578      561      578.000      561.000      3117.00
```

Выход из окна нажатием правой клавиши мышки.

Построим график распределения интенсивности вдоль спектра по 561 строке.

Для создания окна графики служит команда:

```
cre/gra
```

Выводим в окно графики разрез по 561 строке (Y=561):

```
plo/row obj1 561
```






График не виден полностью, так как масштаб по интенсивности в графическом окне (graph) совпадает таковым из окна изображения (display). Масштабируем по интенсивности изображение и снова выводим график:

```
loa/ima obj1 cuts=0
plo/row obj1 561
```



Теперь построим график распределения интенсивности поперек спектра по центру кадра по колонке с номером 600 (X=600). Выводим в окно графики:

```
plo/col obj1 600
```



Командой `get/gcur` (которая работает аналогично `get/cur` только в графическом окне) отметьте две точки недалеко от центрального пика, справа и слева, чтобы определить область между точками для более детального рассмотрения. В терминале появится информация, подобная этой:

```
Frame: obj1
```


X-axis	Y-axis	Pixel	Line	X-position	Y-position	Pixel_value
500.314	1236.98	600	500	600	500	565
616.637	1224.49	600	617	600	617	563

Выход из графического экрана также нажатием правой клавиши мышки. Выведем график между 500 и 617 точками (по сути это колонка с X=600 и Y=500-617):

```
plo/col obj1 600 500,617
```


Видим профиль интенсивности вдоль щели спектрографа.

! Внимание. Когда аргументом команды `Midas` является набор чисел, идущих через запятую (в нашем случае 500,617), они должны быть написаны слитно, без пробелов. Иначе `Midas` воспримет следующее за пробелом число как дополнительный аргумент и результат может быть непредсказуем. Будьте внимательны.




Вычитаем уровень BIAS из кадра с объектом командой и снова посмотрим профиль вдоль щели, замечая, что общий уровень незасвеченной части ПЗС опустился до нуля:

```
comp/ima obj1_b = obj1 - 564
plo/col obj1_b 600
```



Выполняем фильтрацию космических частиц и видим, что большая часть следов удалена:

```
FILTER/COSM obj1_b obj1_cos 0,2.86,0.68
loa/ima obj1_cos cuts=0,0 scale=-3
```




Командой `get/cur` определим координаты углов прямоугольника, включающего

спектр (как показано на предыдущем рисунке), чтобы далее вырезать эту часть кадра и выводим на экран. Нам нужны примерные минимальные и максимальные значения по X и Y (в данном примере X от 60 до 1070, Y от 500 до 630, для опыта перед цифрами поставим знак @, обозначающий, что работаем с пиксельными координатами. В данном случае @ не важен, так как World System Coordinate совпадает с Frame pixel coordinate, но в будущем может пригодиться). Вырезаем нужную часть спектра, исключив по оси X немного края, в которых есть искажения, и включая небольшое пространство по оси Y для дальнейшего учета рассеянного света неба:

```
extr/ima obj1_ext = obj1_cos[@60,@500:@1070,@630]
```

Очищаем окно и смотрим полученный результат, используя при этом уменьшение масштаба в три раза по оси X и увеличение в три раза по оси Y:

```
cle/dis  
loa/ima obj1_ext sc=-3,3 cu=0
```




Те же преобразования выполняем со спектром плоского поля - переводим во внутренний формат, вычитаем уровень BIAS, выделяем такую же область кадра :


```
INDISK/FITS flat1.fits flat1  
comp/ima flat1_b = flat1 - 564  
extr/ima flat1_ext = flat1_b[@60,@500:@1070,@630]
```

Посмотрим как выглядит плоское поле на экране дисплея и на графике (в данном случае @ обязателен, так как системы координат различаются, поскольку мы вырезали изображение):

```
loa/ima flat1_ext sc=-3
```



```
plo/row flat1_ext 561
```



На графике видим существенную неоднородность чувствительности ПЗС матрицы. При выводе на экран мы выводим ту же строку, что и в начале занятия (561). Несмотря на то, что высота вырезанного изображения составляет всего 131 строку, мы выводим 561-ую строку, но в мировых координатах, (начало совпадает с началом исходного кадра ПЗС), что соответствует в пиксельной системе значению @62. Поэтому эквивалентом в пиксельных координатах будет команда `plo/row flat1_ext @62`

Далее нам нужно отнормировать плоское поле так, чтобы среднее значение его интенсивности составило единицу. Смотрим статистику по спектру плоского поля, определяем среднее значение (mean 21934 или median 21483 — они отличаются немного)

```
stat/ima flat1_ext
```


```
frame: flat1_ext (data = R4)
complete area of frame
minimum, maximum:      9.743000e+03      3.239800e+04
at pixel (604,3),(1008,9)
mean, standard_deviation:      2.193386e+04      4.392293e+03
3rd + 4th moment:          0.434023      2.1268
total intensity:          2.90494e+09
exact median, 1. mode, mode:      2.148300e+04      9.787422e+03      1.680603e+04
total no. of bins, binsize:          256      8.884314e+01
# of pixels used = 132441 from 1,1 to 1011,131 (in pixels)
```

и делим спектр на его среднее значение:

```
comp/ima flat1_n = flat1_ext/21934
```

относительное различие неоднородной чувствительности ПЗС:


```
plo/row flat1_n @62
```



Командой `stat/ima flat1_n` мы можем проверить, что действительно среднее значение интенсивности по кадру стало равно почти единице ($9.999937e-01$).

Учитываем неравномерности плоского поля на кадре со спектром объекта, выводим график по 561 строке и видим различие спектра без учета плоского поля и с учетом его:

```
comp/ima obj1_ff = obj1_ext/flat1_n  
loa/ima obj1_ext sc=-3 cu=0  
plo/row obj1_ext 561  
set/gra color=red  
overplo/row obj1_ff 561  
set/gra color=black
```




Видим как изменяется спектр после учета плоского поля (красный цвет).

Теперь основные действия предварительной обработки выполняем со спектром калибровочной лампы (переводим во внутренний формат, вычитаем уровень BIAS, выделяем


такую же область кадра):

```
INDISK/FITS arc1.fits arc1
comp/ima arc1_b = arc1 - 564
loa/ima arc1_b cu=f,max
```




Заметно большое количество следов космических частиц, если их не убрать, то в спектре лампы появится «лишние» линии, которые могут помешать правильному отождествлению. Убираем (параметры те же, что и при обработке кадра с объектом, так как ПЗС матрица та же самая):

```
FILTER/COSM arc1_b arc1_cos 0,2.86,0.68
loa/ima arc1_cos
```



Выделяем область, как и в спектре звезды:

```
extr/ima arc1_ext = arc1_cos[@60,@500:@1070,@630]
loa/ima arc1_ext
```





Переходим непосредственно к обработке спектра. Загружаем пакет long по обработке дисперсионных спектров:

```
set/con long
```


Первым делом мы постараемся учесть рассеянный свет неба. Для этого нам нужно определить участки спектра неба и спектра объекта. Усредняем спектр звезды для определения профиля щели и выводим его:

```
ave/colu obj1_prof = obj1_ff  
plo/row obj1_prof
```



Командой `get/gcur` отметим три пары точек, которые соответствуют границам следующих областей: небо под объектом (на графике слева от пика), небо над объектом (справа от пика) и сам объект (его границей может служить уровень 1-5% от высоты пика).

Например, координаты пар 507,531 595,623 552,568 (как пояснено на следующем рисунке. Снятые вами координаты могут слегка отличаться, это не страшно)




Следующая команда моделирует спектр неба на всем изображении, используя линейную интерполяцию (следующий аргумент после координат неба — степень полинома =1) для каждой колонки (последний аргумент =1) ! Внимание команда использует пиксельные координаты, поэтому вместо 507,531,595,623 запишем 8,32,96,124 и без @ ! :

```
skyfit/long obj1_ff sky 8,32,96,124 1 1
```

Вычитаем спектр неба из спектра звезды, выводим график:


```
comp/ima obj1_sky = obj1_ff - sky
loa/ima obj1_ff cu=0
plo/row obj1_ff 561
set/gra color=red
overplo/row obj1_sky 561
set/gra color=black
```



И можем сделать вывод, что на данном кадре небо не влияет на спектр звезды. Но для других объектов или в других частях спектра влияние может быть замето.


Наконец складываем все строки спектра звезды и получаем одномерный спектр:

```
ave/row obj1_1D = obj1_sky 552,568 SUM
plo/row obj1_1D
```




Приступаем к калибровке по длинам волн над спектром калибровочной лампы. Основные действия мы уже выполнили раньше, осталось просуммировать все строки спектра и получить одномерный спектр:

```
ave/row arc1_1D = arc1_ext 552,568 SUM
plo/row arc1_1D
```



Для калибровки нам нужно найти несколько линии в спектре лампы и отождествить их с имеющимся списком. Сначала оценим параметры для автоматического поиска линий, а именно минимальную интенсивность линий и их характерную толщину. Выведем на графический экран часть спектра лампы (Y=1 — у нас только одна строка, так как спектр одномерный, X от 200 до 400):

```
plo/row arc1_1D 1 200,400
```




Командой `get/gcug` оценим высоту небольших линий (сильно слабые брать не надо) и толщину разных линий. Например, высота около 50, толщина около 10. Поиск эмиссионных линий в спектре лампы выполняется командой:

```
search/long arc1_1D 50 10
```

Можно взглянуть какие линии нашлись (с данными параметрами 38 линий):

```
plo/sear
```



В случае необходимости можно варьировать параметры. Найденные линии записываются в таблицу `line` (файл `line.tbl`). Командой выводится информация о содержимом таблицы:

```
show/tab line
```

Table : line				[Transposed format]	
No.Columns :	8	No.Rows :	38		
All.Columns:	15	All.Rows:	1000	Sel.Rows: 38	
Sorted by :	Sequence	Reference:	Sequence		
Col.#	1:X	Unit:	PIXEL	Format:	F10.2 R*4
Col.#	2:Y	Unit:	PIXEL	Format:	F10.2 R*4
Col.#	3:PEAK	Unit:	DN	Format:	E12.3 R*4
Col.#	4:ERASED	Unit:		Format:	A1 C*4
Col.#	5:IDENT	Unit:	Unitless	Format:	F15.7 R*8
Col.#	6:WAVE	Unit:	Unitless	Format:	F15.7 R*8
Col.#	7:WAVEC	Unit:	Unitless	Format:	F15.7 R*8
Col.#	8:RESIDUAL	Unit:	Unitless	Format:	F15.7 R*8
Selection: ALL					

Следующей командой выводится содержимое таблицы line (можно писать без расширения tbl) по по двум колонкам (к колонкам можно обращаться как по имени, поставив двоеточие, например :X, или по номеру колонки со знаком диез, например #1):


```
read/tab line :X :PEAK
```

Задаем файл с лабораторными длинами волн (MID_ARC: - это специальная директория, где хранятся списки линий калибровочных ламп)

```
set/long LINCAT = MID_ARC:thar25.tbl
```

Начинаем процедуру идентификации нескольких линий, равномерно распределенных по спектру. Отмечаем предлагаемые на рисунке линии (важно, чтобы вертикальная линия курсора попала на линию, то есть по X-координате, высота курсора не играет роли) и для каждой линии пишем длину волны:

```
IDEN/LONG arc1_1D
```



Длины волн отмеченных линий:


3771.37, 3809.46, 3828.38, 3839.70, 3850.58, 3868.52, 3895.42

Midas проводит вычисление дисперсионной кривой $\lambda(X)$ в виде полинома третьей степени, идентификацию найденных линий из таблицы line.tbl с таблицей thar25.tbl, отсеивает линии, которые отскакивают более, чем на 0.05Å, вычисляет окончательную дисперсионную кривую:

```
CALIBRATE/LONG -0.05 3
```

Вывод команды показывает, что использовано 8 линий со средним квадратичным отклонением 0.018 Å. Сам спектр имеет диапазон от 3739.483 Å до 3897.989 Å, дисперсия составила 0.157 Å/pix. Посмотрим остаточные отклонения линий от полинома дисперсионной кривой:

```
plo/resi
```



Сохраняем параметры сессии:


```
save/long my
```

Применяем полученную калибровку к спектру звезды:

```
rebin/long obj1_1D obj1_reb
```


И выводим спектр в длинах волн:

```
plo/row obj1_reb
```




Переходим к учету поглощения. Учитываем поглощение в атмосфере Земли (воздушная масса составила 1.2) и выводим на экран спектры до и после учета поглощения:

```
set/long EXTAB=MID_EXTINCTION:atmoexan
EXTINC/LONG obj1_reb obj1_air 1.2
cuts/ima obj1_reb 0,30000
plo/row obj1_reb
set/gra color=red
overplo/row obj1_air
set/gra color=black
```



Учитываем межзвездное поглощение с покраснением $E(B-V)=0.5^m$ и выводим на экран окончательный спектр по сравнению с предыдущим:

```
set/long EXTAB=MID_EXTINCTION:instexan
EXTINC/LONG obj1_red obj1_red 0.5
cuts/ima obj1_red 0,210000
plo/row obj1_red
set/gra color=green
overplo/row obj1_red
set/gra color=black
```



Сохраняем окончательный спектр в FITS-формате для анализа в других программах:

```
outdisk/fits obj1_red obj1_red.fits
```